Pacific Southwest
Forest & Range
Experiment Station

P. O. BOX 245
BERKELEY, CALIFORNIA 94701

1968

U.S.D.A.
Forest Service
Research Note
PSW-180

Delineating Landscape View Areas . . .

a computer approach

Elliot L. Amidon Gary H. Elsner

ABSTRACT: The terrain visible from a given point can be determined quickly and efficiently by a computer. A FORTRAN subprogram--called VIEWIT--has been developed for this purpose. Input consists of data on elevations, by coordinates, which can be obtained from maps or aerial photos. The computer will produce an overlay that shows the maximum area visible from an observation point. Both the observation point and the length of line-of-sight are variable. This new technique makes it unnecessary to draw overlays manually.

RETRIEVAL TERMS: view areas; scene area analysis; computer programs.

OXFORD: 907.1:U712.2--U681.3.

Whether viewed from an observation point or a moving vehicle, a forest landscape has distinctive esthetic value. Wildland managers have begun to delineate the boundaries of views as a step toward protecting scenic values. Delineating maximum visible areas can have other benefits, as for example, in evaluating alternative sites for locating fire towers or scenic overlooks.

To identify view areas, the forester must go out in the field. He can outline a boundary on a map, as seen from one or more vantage points. Boundaries will vary--depending on the observer's particular position, surrounding topography, visibility, and the observer himself. Field work will always be required to determine final boundaries. But a map overlay displaying the maximum area visible for the particular topography would reduce field work.

An overlay can be drawn manually by constructing terrain profiles.[1] The cost of constructing hundreds of profiles would, however, be prohibitive. The military services have encountered similar problems in calculating line-of-sight and low-altitude flight paths. Our review of military algorithms suggested that a digital computer program would offer an economically feasible procedure for delineating view areas.[2] But existing computer programs were not directly suited for delineating visible areas. A new program was called for.

We have developed a FORTRAN subprogram--called VIEWIT--to compute the area visible from an observer's position. Computations of observations of 1-1/2-mile radius required only 1 second on a UNIVAC 1108 computer (fig. 1). The radius of the view area can be varied to allow for analyzing different size areas from the same vantage point. Computation time using the new procedure was 50 times faster than that with other computer programs. And it is no longer necessary to draw overlays manually.

Forest Service
U.S. Department of Agriculture

(Reaffirmed and reprinted, 1976. See also General Technical Report PSW-11 and Research Paper PSW-91 issued by this Station.)



Figure 1.--*Area visible within 1-1/2 miles of the observer's position for the hypothetical data shown in Figure 2. The number 11 designates visible points.*



Figure 2.--*Hypothetical data on terrain, with elevations shown in hundreds of feet. They are the input to the VIEWIT subprogram for determining areas visible from given points.*

The basic data required to compute visible areas--elevations, by coordinates--are simple but voluminous (fig. 2). They may be read directly from topographic maps or interpreted stereoscopically from aerial photographs. For a given data matrix of elevations and radius, VIEWIT will compute the visible area from a specified observation point. The user must write a mainline program to call VIEWIT and to specify input and output for his particular data set.

The complete VIEWIT subprogram is as follows:

```
@I FOR VIEWIT,VIEWIT
      SUBROUTINE VIEWIT(SCALIN,SIZEI,SIZEJ,RADIUS,IOBS,JOBS,KARDS,LEVEL,
     1INRIM,MDIM)
C VIEWIT SELECTS THE CELLS LYING ON A CIRCLE OF ANY GIVEN RADIUS WITH
C        CENTER AT IOBS,JOBS.
C IOBS   I COORDINATE OF OBSERVER
C JOBS   J COORDINATE OF OBSERVER
C INRIM  OUTPUT MATRIX. WHEN CELL I,J IS OUTSIDE VISIBILITY RANGE,
C        INRIM(I,J)=88 . WHEN I,J IS VISIBLE FROM IOBS,JOBS, INRIM =11 .
C        WHEN I,J IS NOT VISIBLE, INRIM = 0 .
C LEVEL  INPUT MATRIX OF ELEVATIONS IN HUNDREDS OF FEET .
C KF     I DIMENSION DIRECTION COEFFICIENT . WHEN KF=-1 , I IS DECREMENT
C        ED FROM IOBS . WHEN KF = 1 , I IS INCREMENTED FROM IOBS .
C KS(I)  J DIMENSION DIRECTION COEFFICIENT. KS(1) DECREMENTS J FROM JOBS,
C        KS(2) INCREMENTS J FROM JOBS
C SCALIN RECIPROCAL OF MAP REPRESENTATIVE FRACTION,E.G.,''31680''.
C SIZEI NO. OF CELLS PER INCH, MEASURED HORIZONTALLY
C SIZEJ NO. OF CELLS PER INCH, MEASURED VERTICALLY
C RADIUS RANGE OF VISIBILITY IN MILES
C JRADIM RADIUS IN J CELL DIMENSION .
C RADIM  SQUARE OF THE RADIUS (HYPOTENUSE OF RT TRIANGLE WHOSE VERTICES
C        ARE (IOBS,JOBS), (IOBS,JM),(I,JM) .
C JM     DECREMENTED BASE OF TRIANGLE IN J CELL DIMENSION
C RM     DECREMENTED BASE OF TRIANGLE IN MILES .
C TLL    LEFT-MOST CELL IN LINE OF SELLS THROUGH WHICH CIRCLE PASSES TO
C        LEFT OF IOBS , AT A GIVEN RM.
C IL     RIGHT-MOST CELL IN THAT LINE .
C IML    LEFT - MOST CELL IN THAT LINE , TO RIGHT OF IOBS
C IM     RIGHT- MOST CELL IN THAT LINE , TO RIGHT OF IOBS
      DIMENSION KS(2),INRIM(MDIM,1),LEVEL(MDIM,1)
      KS(1)=-1
      KS(2)=1
      SCALIN=63360/SCALIN
      JRADIM =SIZEJ * SCALIN * RADIUS
      SIZEI= SIZEI*SCALIN
      SIZEJ = SIZEJ*SCALIN
      RADIM = RADIUS**2
      JMIN = JOBS-JRADIM
      IL=IOBS-1
      IM=IOBS+1
      RM=RADIUS
      JM=RM*SIZEJ
C     FIRST SPECIAL CASE,PERFORMED ONCE,AVOIDING DIVISION BY ZERO
      M=IOBS
      KF=0
      DO 22 L=1,2
      N=JOBS+KS(L)*JM
      CALL GENRAY(IOBS,JOBS,M,N,KF,L,KS,INRIM,LEVEL,MDIM)
   22 CONTINUE
      JBEG=JMIN+1
      DO 30 JY=JBEG,JOBS
      JM=JM-1
      RM = RM - 1./SIZEJ
      RJ = RM**2
```

```
      RADCAL=RADIM-RJ
      RIM=SQRT(RADCAL)
      LIM = RIM*SIZEI
      ILL = IL
      IL = IOBS - LIM
      IML = IM
      IM=IOBS+LIM
      DO 28 L=1,2
      N=JOBS+KS(L)*JM
      KF=-1
      DO 24 M=IL,ILL
      CALL GENRAY(IOBS,JOBS,M,N,KF,L,KS,INRIM,LEVEL,MDIM)
   24 CONTINUE
      KF=1
      DO 26 M=IML,IM
      CALL GENRAY(IOBS,JOBS,M,N,KF,L,KS,INRIM,LEVEL,MDIM)
   26 CONTINUE
   28 CONTINUE
   30 CONTINUE
      RETURN
      END
@I FOR GENRAY,GENRAY
      SUBROUTINE GENRAY(IOBS,JOBS,I,J,KF,L,KS,INRIM,LEVEL,MDIM)
C GENRAY GENERATES A RAY OF CELLS FROM IOBS,JOBS TO EACH I,J ON THE
C      CIRCLE . THE RAY CONSISTS OF ALL AND ONLY THOSE CELLS THROUGH
C        WHICH A LINE FROM THE OBSERVER TO A RIM POINT PASSES .
C ANGMAX MAXIMAL ANGLE BETWEEN OBSERVER AND ANY CELL ON THE RAY WHICH
C        HAS BEEN TESTED FOR VISIBILITY .
C USER MUST CHOOSE INITIAL VALUE OF ANGMAX. SELECT ZERO IF THE CENTER
C CELL IS TREATED AS A PLANE WITH THE OBSERVER FIXED AT ITS CENTER. ALL
C TERRAIN AT A LESSER ELEVATION WILL BE INVISIBLE. OTHERWISE SELECT A
C LARGE NEGATIVE VALUE(EG,-999) WHICH TREATS THE CELL AS A POINT.
C JLEG   NUMBER OF CELLS BETWEEN JOBS AND J .
C ILEG   NUMBER OF CELLS BETWEEN IOBS AND I .
      DIMENSION KS(2),INRIM(MDIM,1),LEVEL(MDIM,1)
      ANGMAX=-999
      JLEG=IABS(J-JOBS)
      IF(IOBS.EQ.I) GO TO 35
      ILEG=IABS(I-IOBS)
      IF(ILEG.LT.JLEG) GO TO 20
      I=IOBS
      DO 10 K=1,ILEG
      I=I+KF
      J =JOBS+KS(L)*JLEG*K/ILEG
      RSPAN=IABS(IOBS-I)
      CALL VISTRY(I,J,IOBS,ANGMAX,INRIM,LEVEL,RSPAN,JOBS,MDIM)
   10 CONTINUE
      GO TO 40
   20 J=JOBS
      DO 30 K=1,JLEG
      J=J+KS(L)
      I =IOBS+KF*ILEG*K/JLEG
      RSPAN=IABS(JOBS-J)
      CALL VISTRY(I,J,IOBS,ANGMAX,INRIM,LEVEL,RSPAN,JOBS,MDIM)
   30 CONTINUE
      GO TO 40
C     SPECIAL CASE (VERTICAL RAY) AVOIDING DIVISION BY ZERO
   35 KF=KS(L)
      J=JOBS
      DO 38 K=1,JLEG
      J=J+KF
      RSPAN=IABS(JOBS-J)
      CALL VISTRY(I,J,IOBS,ANGMAX,INRIM,LEVEL,RSPAN,JOBS,MDIM)
   38 CONTINUE
   40 RETURN
      END
```

```
aI FOR VISTRY.VISTRY
      SUBROUTINE VISTRY(I,J,IOBS,ANGMAX,INRIM,LEVEL,RSPAN,JOBS,MDIM)
C VISTRY DETERMINES VISIBILITY OF SUCCESSIVE CELLS ALONG THE RAY .
C ANGNEX ANGLE BETWEEN OBSERVER AND I,J .
      DIMENSION INRIM(MDIM,1),LEVEL(MDIM,1)
      ANGNEX=(LEVEL(I,J)-LEVEL(IOBS,JOBS))/RSPAN
      IF(ANGNEX.GE.ANGMAX) GO TO 10
      INRIM(I,J)=0
      GO TO 20
   10 INRIM(I,J)=11
      ANGMAX=ANGNEX
   20 RETURN
      END
```

FOOTNOTES

[1]Spurr, Stephen H. *Photogrammetry and photo-interpretation*. Ed. 2. New York: Ronald Press Co. 472 p., illus. 1960.

[2]Bardoll, Ivan H., III *Computerized visibility calculations maximum sighting range program*. San Diego: Univ. of Calif. 121 p. Jul. 1967.

Boehne, R. C. *Digital simulation of environmental limitations of intervisibility*. *In*, Digital simulation in operational research. S.H. Hollingdale, ed. New York: American Elsevier Publ. Co. p. 227-238, illus. 1967.

Ford, Lester R., Jr., Isaacson, H. S., and Pethel, F. C. *Computer terrain simulation for line-of-sight calculations*. Oper. Res. J. 4:478-482, illus. 1959.

McGrath, James J., ed. *Aeronautical charts and map display*. Goleta, Calif.: Human Factors Research, Inc. 237 p., illus. Aug. 1967.

The Authors

ELLIOT L. AMIDON, an economist, is conducting forest economics studies, with particular emphasis on multiple-use problems. He earned B.S. (1954), in forest management at Colorado Agricultural and Mechanical College, and M.S. degrees (1960), in agricultural economics at the University of California. He has been with the Berkeley station staff since 1959. GARY H. ELSNER is studying quantitative planning problems in forest recreation. He earned a B.S. degree (1962) in agricultural economics at the University of Arkansas, and M.S. (1964) and Ph.D. (1966) degrees at the University of California, Berkeley.